
Mplus Model Code for the Single and Multiple Source Latent State Trait Analyses

	These supplemental materials first show the Mplus model code for the single source latent state trait model with the mother as the source. The code for the other three single source latent state trait models (i.e., for fathers, primary teachers, and secondary teachers) is the same. The supplemental materials then show the Mplus code for the multiple source (mothers, fathers, primary teachers, and secondary teachers with mothers as the reference source) latent state trait model.

1.  Mplus Model Code for the Single Source Latent State Trait Analysis.  This code corresponds to Figure 1 in the paper.  

! Key:  T = trait factor; O = state (occasion-specific) residual factor; M = mother; t = time (two occasions of measurement); p = parcel (three parcels).

Variable:

Names are Mt1p1 Mt1p2 Mt1p3 Mt2p1 Mt2p2 Mt2p3; !parcels for mothers 

Analysis:  Estimator = MLR;

Model:

!  Three parcel specific trait factors with the loadings set to 1 

T1_M by Mt1p1@1 
                Mt2p1@1;
T2_M by Mt1p2@1 
                Mt2p2@1; 
T3_M by Mt1p3@1 
                Mt2p3@1;

! Two state residual factors with loadings on parcels 2 and 3 are held equal across occasions 

O1_M by Mt1p1@1 
                Mt1p2 (gamma2)
                Mt1p3 (gamma3);

O2_M by Mt2p1@1
                Mt2p2 (gamma2)
                Mt2p3 (gamma3);

! Trait and state residual factor correlations are set to 0 due to model requirements

T1_M-T3_M with O1_M-O2_M@0;

! State residual factor correlation set to 0 due to due to model requirements 

O1_M with O2_M@0;

! Means of trait factors 

[T1_M T2_M T3_M];

! Intercepts of parcels set to 0 to identify factor means 

[Mt1p1@0 Mt1p2@0 Mt1p3@0 Mt2p1@0 Mt2p2@0 Mt2p3@0];


2.  Mplus Model Code for the Multiple Source Latent State Trait Analysis.  This code corresponds to Figure 2 in the paper.

! Key:  T = reference trait factors that are defined by mother reports and are parcel-specific across sources; TS = residual trait factors that pertain to the non-reference sources (fathers, primary teachers, and secondary teachers) and are source- and parcel-specific; O = reference state residual factors that are defined by mother reports and are parcel-specific across sources; OS = state residual factors that pertain to the non-reference sources (fathers, primary teachers, and secondary teachers) and are source-specific across parcels. M = mother; F = Father; TA = primary teacher; TB = secondary teacher; t = time (two occasions of measurement); p = parcel (three parcels).

Variable:
  
Names are  	Mt1p1 Mt1p2 Mt1p3 Mt2p1 Mt2p2 Mt2p3            !parcels for mothers
  		Ft1p1 Ft1p2 Ft1p3 Ft2p1 Ft2p2 Ft2p3                    !parcels for fathers                  
  		TAt1p1 TAt1p2 TAt1p3 TAt2p1 TAt2p2 TAt2p3  !parcels for primary teachers
  		TBt1p1 TBt1p2 TBt1p3 TBt2p1 TBt2p2 TBt2p3;  !parcels for secondary teachers

Analysis: Estimator = MLR;

Model:

!  Three parcel specific reference trait factors with mothers (M) as the reference source.  
! Factor loadings are held equal across time for all sources 
 
   T1_M by                Mt1p1@1 
                                  Mt2p1@1
    		          Ft1p1 (lambda1F)
    		          Ft2p1 (lambda1F)
    		           
    		          TAt1p1 (lambda1A)
    		          TAt2p1 (lambda1A)
    		          
     		          TBt1p1 (lambda1B)
    		          TBt2p1 (lambda1B); 
    		
    T2_M by                Mt1p2@1 
                                   Mt2p2@1
    		           Ft1p2 (lambda2F)
    		           Ft2p2 (lambda2F)
    		           
    		           TAt1p2 (lambda2A)
    		           TAt2p2 (lambda2A)
    		           
     		           TBt1p2 (lambda2B)
    		           TBt2p2 (lambda2B); 
    		
    T3_M by                Mt1p3@1 
                                   Mt2p3@1
    		           Ft1p3 (lambda3F)
    		           Ft2p3 (lambda3F)
    		           
    		           TAt1p3 (lambda3A)
    		           TAt2p3 (lambda3A)
    		           
     		           TBt1p3 (lambda3B)
    		           TBt2p3 (lambda3B); 

! Trait residual factors for fathers (F), primary teachers (TA) and secondary teachers (TB) 

! Factor loadings are held equal across time for all sources
 
      TS_F1 by    Ft1p1@1 
                          Ft2p1@1;

      TS_F2 by    Ft1p2@1 
                          Ft2p2@1;
      
      TS_F3 by     Ft1p3@1 
                           Ft2p3@1; 

      TS_TA1 by   TAt1p1@1 
                            TAt2p1@1;

      TS_TA2 by   TAt1p2@1 
                            TAt2p2@1;

      TS_TA3 by   TAt1p3@1 
                            TAt2p3@1; 

! The next four line of codes indicate that a single residual trait factor was sufficient for TB (secondary teachers)

       TS_TB by  TBt1p1@1 
                          TBt2p1@1
                          TBt1p2 TBt2p2  (TB_P2)
                          TBt1p3 TBt2p3  (TB_P3);  

! Occasion (state) residual factors (parcel specific) with mothers as reference source

! Factor loadings are set equal across time 

!  Occasion 1, Parcel 1
    
O1_M1 by Mt1p1@1
    	       Ft1p1     (delta1F)
    	       TAt1p1  (delta1A)
    	       TBt1p1;  (delta1B);

!  Occasion 1, Parcel 2

O1_M2 by Mt1p2@1
    	       Ft1p2     (delta2F)
    	       TAt1p2   (delta2A)
    	       TBt1p2   (delta2B);
	
!  Occasion 1, Parcel 3

O1_M3 by Mt1p3@1
    	       Ft1p3     (delta3F)
    	       TAt1p3  (delta3A)
    	       TBt1p3  (delta3B); 	

!  Occasion 2, Parcel 1

O2_M1 by Mt2p1@1
    	       Ft2p1     (delta1F)
    	       TAt2p1  (delta1A)
    	       TBt2p1  (delta1B); 

!  Occasion 2, Parcel 2

O2_M2 by Mt2p2@1
                   Ft2p2      (delta2F)
    	       TAt2p2   (delta2A)
    	       TBt2p2  (delta2B); 

!  Occasion 2, Parcel 3
	
O2_M3 by Mt2p3@1
    	       Ft2p3      (delta3F)
    	       TAt2p3   (delta3A)
    	       TBt2p3   (delta3B);	

!  Occasion specific source residual factors (fathers, primary teaches, and secondary teachers)

! Factor loadings are set equal across time 

!  Occasion 1, parcels for fathers
    
OS_Ft1 by Ft1p1@1
      	      Ft1p2 (gamma2F)
    	      Ft1p3 (gamma3F);

 !  Occasion 2, parcels for fathers
  
OS_Ft2 by Ft2p1@1
      	      Ft2p2 (gamma2F)
    	      Ft2p3 (gamma3F); 

 !  Occasion 1, parcels for primary teachers
  	
 OS_TAt1 by TAt1p1@1
     	           TAt1p2 (gamma2TA)
    	           TAt1p3 (gamma3TA); 

!  Occasion 2, parcels for primary teachers
    	
OS_TAt2 by TAt2p1@1
    	          TAt2p2 (gamma2TA)
    	          TAt2p3 (gamma3TA);  	

!  Occasion 1, parcels for secondary teachers    	

OS_TBt1 by TBt1p1@1
    	          TBt1p2 (gamma2TB)
    	          TBt1p3 (gamma3TB);

!  Occasion 2, parcels for secondary teachers    	
    
OS_TBt2 by TBt2p1@1
    	         TBt2p2 (gamma2TB)
    	         TBt2p3 (gamma3TB);	
    	
 !  Estimate latent trait means of the reference trait factors

    [T1_M-T3_M*];

!  Reference trait factors are allowed to correlate 

    T1_M-T3_M WITH T1_M-T3_M*;

!  Reference occasion residual factors for the same occasion are allowed to correlate 

    O1_M1-O1_M3 WITH O1_M1-O1_M3; !Occasion 1 factors
    O2_M1-O2_M3 WITH O2_M1-O2_M3; !Occasion 2 factors

!  All trait residual factors are allowed to correlate  

    TS_F1-TS_F3 WITH TS_F1-TS_F3* TS_TA1-TS_TA3* TS_TB*;

    TS_TA1-TS_TA3 WITH TS_F1-TS_F3* TS_TA1-TS_TA3* TS_TB*;

    TS_TB WITH TS_F1-TS_F3* TS_TA1-TS_TA3*;

!  Occasion residual factors for the same occasion are allowed to correlate 

!  Occasion 1 residual factors

    OS_Ft1 OS_TAt1 OS_TBt1 with OS_Ft1 OS_TAt1 OS_TBt1; 

!  Occasion 2 residual factors

    OS_Ft2 OS_TAt2 OS_TBt2 with OS_Ft2 OS_TAt2 OS_TBt2; 

! Reference occasion residual factors are not allowed to correlate across time

    O1_M1 O2_M1 with O1_M1@0 O2_M1@0 ;
    O1_M2 O2_M2 with O1_M2@0 O2_M2@0 ;
    O1_M3 O2_M3 with O1_M3@0 O2_M3@0 ;

!  Occasion residual factors are not allowed to correlate across time

    OS_Ft1 OS_Ft2 with OS_Ft1@0 OS_Ft2@0;
    OS_TAt1 OS_TAt2 with OS_TAt1@0 OS_TAt2@0;
    OS_TBt1 OS_TBt2 with OS_TBt1@0 OS_TBt2@0; 

!  Error variances are set equal across time for the same parcels 

    Mt1p1 Mt2p1    (epsM1);
    Mt1p2 Mt2p2    (epsM2);
    Mt1p3 Mt2p3    (epsM3);
    Ft1p1 Ft2p1        (epsF1);
    Ft1p2 Ft2p2        (epsF2);
    Ft1p3 Ft2p3        (epsF3);
    TAt1p1 TAt2p1  (epsTA1);
    TAt1p2 TAt2p2  (epsTA2);
    TAt1p3 TAt2p3  (epsTA3);
    TBt1p1 TBt3p1  (epsTB1);
    TBt1p2 TBt3p2  (epsTB2);
    TBt1p3 TBt3p3  (epsTB3);

!  Intercepts for mothers are set to 0 to identify factor means

    [Mt1p1@0 Mt1p2@0 Mt1p3@0];
    [Mt2p1@0 Mt2p2@0 Mt2p3@0];
     	
!  Estimate intercepts for fathers (set equal for same parcel over time)

     [Ft1p1 Ft2p1] (intF1);
     [Ft1p2 Ft2p2] (intF2);
     [Ft1p3 Ft2p3] (intF3);

!  Estimate intercepts for primary teachers (set equal for same parcel over time) 

     [TAt1p1 TAt3p1] (intTA1);
     [TAt1p2 TAt3p2] (intTA2);
     [TAt1p3 TAt3p3] (intTA3);

!  Estimate intercepts for secondary teachers (set equal for same parcel over time) 

     [TBt1p1 TBt3p1] (intTB1);
     [TBt1p2 TBt3p2] (intTB2);
     [TBt1p3 TBt3p3] (intTB3);
   

